Файл Tester.Frm

' ДЛЯ ВКЛЮЧЕНИЯ ОТЛАДКИ РАСКОММЕНТАРИТЬ

' СТРОКИ В КОТОРЫХ ЕСТЬ СЛОВО debug

'

'

Private Sub Cub(X, y, R, V1, V2, V3) ' Кубик

 Dim Dx, Dy As Integer

 Dx = 2 * R / 3

 Dy = R / 4

 If V1 = 1 Then 'Передняя грань

  Line (X, y)-Step(R, R), , B

 End If

 If V2 = 1 Then 'Верхняя грань

  Line (X, y)-Step(R, 0)

  Line (X, y)-Step(Dx, -Dy)

  Line (X + R, y)-Step(Dx, -Dy)

  Line (X + Dx, y - Dy)-Step(R, 0)

 End If

 If V3 = 1 Then 'Боковая грань

  Line (X + R, y)-Step(Dx, -Dy)

  Line (X + R, y + R)-Step(Dx, -Dy)

  Line (X + R, y)-Step(0, R)

  Line (X + R + Dx, y - Dy)-Step(0, R)

 End If

End Sub

Private Function Swap3(A, B, C As Double)

' Меняем местами 3 числа случайным образом - для вывода ответов в случайном порядке

'При вызове число A содержит правильный ответ. Возвращает новую позицию числа А (1,2,3)

 Dim X As Double

 Dim Rul As Integer

 Rul = 1

 If Rand(2) = 0 Then

  X = A

  A = B

  B = X

  Rul = 2

 End If

 If Rand(2) = 0 Then

  X = B

  B = C

  If Rul = 2 Then Rul = 3

  C = X

 End If

 If Rand(2) = 0 Then

  X = A

  If Rul = 2 Then

   Rul = 1

   GoTo Line100

  End If

  If Rul = 1 Then Rul = 2

Line100:

  A = B

  B = X

 End If

 Swap3 = Rul

End Function

Private Function Rand(N) 'Случайное целое от 0 до N-1

 Rand = Int(N * Rnd)

End Function

Private Sub MakePirr(x1, y1, x2, y2, x3, y3, x4, y4) '3-угольная пирамида

 Line (x1, y1)-(x2, y2)

 Line (x2, y2)-(x3, y3)

 Line (x3, y3)-(x4, y4)

 Line (x4, y4)-(x1, y1)

 Line (x1, y1)-(x3, y3)

 DrawStyle = 1

 Line (x2, y2)-(x4, y4)

 DrawStyle = 0

End Sub

Private Function Rast(x1, y1, x2, y2 As Double) 'Расстояние между точками

 Rast = Sqr((x1 - x2) ^ 2 + (y1 - y2) ^ 2)

End Function

Private Function UrLine(X, x1, y1, x2, y2 As Double) 'Уравнение прямой по 2 точкам

 UrLine = ((X - x1) / (x2 - x1)) * (y2 - y1) + x1

End Function

Private Function Otn(x1, x2, L As Double)

'Делит отрезок от x1 до x2 в отношении L, L<>-1

 Otn = (x1 + L * x2) / (1 + L)

End Function

Private Function Min(x1, x2) 'Минимум из 2 чисел

 If x1 < x2 Then Min = x1 Else Min = x2

End Function

Private Sub Triangle(x1, y1, x2, y2, x3, y3)

'Треугольник по координатам 3 сторон

 Line (x1, y1)-(x2, y2)

 Line (x2, y2)-(x3, y3)

 Line (x1, y1)-(x3, y3)

End Sub

Private Sub MakePirr4(x1, y1, x2, y2, x3, y3, x4, y4, x5, y5) '4-угольная пирамида

 Line (x1, y1)-(x2, y2)

 Line (x1, y1)-(x3, y3)

 Line (x1, y1)-(x4, y4)

 Line (x2, y2)-(x3, y3)

 Line (x3, y3)-(x4, y4)

 DrawStyle = 2

 Line (x2, y2)-(x5, y5)

 Line (x4, y4)-(x5, y5)

 Line (x1, y1)-(x5, y5)

 DrawStyle = 0

End Sub

Private Sub Pirr(N, Sx, Sy) 'Разные приамиды для вопроса 5

 Dim X, y As Double

 Scal = 10

 If N = 1 Then 'Правильный ответ

  Call MakePirr(Sx, Sy, Sx - 3 * Scal, Sy + 9 * Scal, _

   Sx - Scal, Sy + 11 * Scal, Sx + 4 * Scal, Sy + 9 * Scal)

  CurrentX = Sx - 3 * Scal

  CurrentY = Sy + 10 * Scal

  Print "a"

  X = Otn(Sx, Sx + 4 * Scal, 1.7)

  y = Otn(Sy, Sy + 9 * Scal, 1.7)

  DrawStyle = 2

  Line (Sx - 3 * Scal, Sy + 9 * Scal)-(X, y)

  Line (Sx - Scal, Sy + 11 * Scal)-(X, y)

  x1 = Otn(Sx - 3 * Scal, X, 3)

  y1 = Otn(Sy + 9 * Scal, y, 3)

  x2 = Otn(Sx - Scal, X, 3)

  y2 = Otn(Sy + 11 * Scal, y, 3)

  DrawStyle = 0

  Line (x1, y1)-(x2, y2)

  CurrentX = X + 10

  CurrentY = y - 10

  Print "d"

 ElseIf N = 2 Then 'Угол при вершине

  Call MakePirr(Sx, Sy, Sx - 3 * Scal, Sy + 9 * Scal, Sx + Scal, _

   Sy + 11 * Scal, Sx + 4 * Scal, Sy + 9 * Scal)

  CurrentX = Sx - 2 * Scal

  CurrentY = Sy + 5 * Scal

  Print "a"

  X = Otn(Sx, Sx + 4 * Scal, 1.7)

  y = Otn(Sy, Sy + 9 * Scal, 1.7)

  DrawStyle = 2

  Line (Sx - 3 * Scal, Sy + 9 * Scal)-(X, y)

  Line (Sx + Scal, Sy + 11 * Scal)-(X, y)

  x1 = Otn(Sx - 3 * Scal, X, 3)

  y1 = Otn(Sy + 9 * Scal, y, 3)

  x2 = Otn(Sx + Scal, X, 3)

  y2 = Otn(Sy + 11 * Scal, y, 3)

  DrawStyle = 0

  Line (x1, y1)-(x2, y2)

  CurrentX = X + 4

  CurrentY = y - 10

  Print "d"

 ElseIf N = 3 Then 'С боковой стороной A

  Call MakePirr(Sx, Sy, Sx - 3 * Scal, Sy + 9 * Scal, Sx - Scal, _

   Sy + 11 * Scal, Sx + 4 * Scal, Sy + 9 * Scal)

  CurrentX = Sx - 3 * Scal

  CurrentY = Sy + 10 * Scal

  Print "a"

  x1 = Otn(Sx, Sx - 3 * Scal, 0.2)

  y1 = Otn(Sy, Sy + 9 * Scal, 0.2)

  x2 = Otn(Sx, Sx + 4 * Scal, 0.2)

  y2 = Otn(Sy, Sy + 9 * Scal, 0.2)

  DrawStyle = 2

  Line (x1, y1)-(x2, y2)

  DrawStyle = 0

  CurrentX = Sx + 8

  CurrentY = Sy

  Print "d"

 ElseIf N = 4 Then 'Треугольные основания

  Sx = Sx - 3 * Scal

  Call Triangle(Sx, Sy, Sx + 6 * Scal, Sy, Sx + 3 * Scal, Sy + 2 * Scal)

  Line (Sx, Sy + 10 * Scal)-(Sx + 3 * Scal, Sy + 12 * Scal)

  Line (Sx + 3 * Scal, Sy + 12 * Scal)-(Sx + 6 * Scal, Sy + 10 * Scal)

  Line (Sx, Sy)-(Sx, Sy + 10 * Scal)

  Line (Sx + 3 * Scal, Sy + 2 * Scal)-(Sx + 3 * Scal, Sy + 12 * Scal)

  Line (Sx + 6 * Scal, Sy)-(Sx + 6 * Scal, Sy + 10 * Scal)

  y1 = Otn(Sy, Sy + 10 * Scal, 1)

  Line (Sx + 3 * Scal, Sy + 12 * Scal)-(Sx + 6 * Scal, y1)

  DrawStyle = 2

  Line (Sx, Sy + 10 * Scal)-(Sx + 6 * Scal, Sy + 10 * Scal)

  Line (Sx + 6 * Scal, y1)-(Sx, Sy + 10 * Scal)

  x2 = Otn(Sx, Sx + 6 * Scal, 3)

  y2 = Otn(Sy + 10 * Scal, y1, 3)

  x1 = Otn(Sx + 3 * Scal, Sx + 6 * Scal, 3)

  y1 = Otn(Sy + 12 * Scal, y1, 3)

  Line (x1, y1)-(x2, y2)

  DrawStyle = 0

  CurrentX = Sx + Scal

  CurrentY = Sy + 11 * Scal

  Print "a"

  CurrentX = Sx + 6 * Scal

  CurrentY = Sy + 5 * Scal

  Print "d"

 ElseIf N = 5 Then 'В основании параллелограмм

  Call MakePirr4(Sx, Sy, Sx - 4 * Scal, Sy + 11 * Scal, Sx + 3 * Scal, _

   Sy + 11 * Scal, Sx + 5 * Scal, Sy + 8 * Scal, Sx - 2 * Scal, Sy + 8 * Scal)

  CurrentX = Sx + Scal

  CurrentY = Sy + 11 * Scal

  Print "a"

  DrawStyle = 2

  Line (Sx - 4 * Scal, Sy + 11 * Scal)-(Sx + 5 * Scal, Sy + 8 * Scal)

  DrawStyle = 0

  CurrentX = Sx - 4 * Scal

  CurrentY = Sy + 11 * Scal

  Print "d"

 ElseIf N = 6 Then 'Угол находится в основании

  Call MakePirr(Sx, Sy, Sx - 3 * Scal, Sy + 9 * Scal, Sx - Scal, _

   Sy + 11 * Scal, Sx + 4 * Scal, Sy + 9 * Scal)

  CurrentX = Sx - 3 * Scal

  CurrentY = Sy + 10 * Scal

  Print "a"

  x1 = Otn(Sx - 3 * Scal, Sx + 4 * Scal, 2)

  x2 = Otn(Sx - Scal, Sx + 4 * Scal, 2)

  y2 = Otn(Sy + 11 * Scal, Sy + 9 * Scal, 2)

  DrawStyle = 0

  Line (x1, Sy + 9 * Scal)-(x2, y2)

  CurrentX = Sx + 3 * Scal

  CurrentY = Sy + 9 * Scal

  Print "d"

 End If

End Sub

Private Sub Fig(Sx0, Sy0, R, Height1, Depth1, Step1, Step2, Step3, Add1)

'Рисует фигуры для вопросов 1 и 10

 V1 = 1

 V2 = 1

 V3 = 1

 Sx = Sx0

 Sy = Sy0

 For k = 0 To Depth1 - 1

  If k = Depth1 - 1 Then V1 = 1 Else V1 = 0

  For i = 0 To Step1 - 1

   If i = Step1 - 1 Then V3 = 1 Else V3 = 0

   If i = Step1 - 1 Then

    If k = 0 Then V3 = 0

   End If

   For j = 0 To Height1 - 1

    If j = 0 Then V2 = 1 Else V2 = 0

    If i = Step1 - 1 Then

     If j = Height1 - 1 Then V3 = 0

    End If

    Call Cub(Sx + i * R, Sy + j * R, R, V1, V2, V3)

   Next j

  Next i

  Sx = Sx - 2 * R / 3

  Sy = Sy + R / 4

 Next k

 'Вторая секция - низ

 Sx = Sx0 - (Depth1 - 1) * (2 * R / 3) + Step1 * R

 Sy = Sy0 + (Depth1 - 1) * (R / 4) + (Height1 - 1) * R

 For k = 0 To Depth1 - 1

  For i = 0 To Step2 - 1

   If i = Step2 - 1 Then V3 = 1 Else V3 = 0

   If k = 0 Then V1 = 1 Else V1 = 0

   If k = Depth1 - 1 Then V2 = 0 Else V2 = 1

   If k = Depth1 - 1 Then V3 = 0

   Call Cub(Sx + i * R, Sy, R, V1, V2, V3)

  Next i

  Sy = Sy - R / 4

  Sx = Sx + 2 * R / 3

 Next k

 'Третья секция - верх, заканчивает вырез

 Sx = Sx0 + Step1 * R

 Sy = Sy0

 For i = 0 To Step2 - 1

  For j = 0 To Height1 - 1

   If j = 0 Then V2 = 1 Else V2 = 0

   If j = Height1 - 1 Then V1 = 0 Else V1 = 1

   Call Cub(Sx + i * R, Sy + j * R, R, V1, V2, 0)

  Next j

 Next i

 '4-я секция - линейная

 Sx = Sx0 + R * (Step1 + Step2)

 Sy = Sy0

 For i = 0 To Step3 - 1

  If i = Step3 - 1 Then V3 = 1 Else V3 = 0

  For j = 0 To Height1 - 1

   If j = 0 Then V2 = 1 Else V2 = 0

   If j = Height1 - 1 Then V3 = 0

   Call Cub(Sx + i * R, Sy + j * R, R, 1, V2, V3)

  Next j

 Next i

 'Добавляем или нет еще 1 кубик сбоку

 Sx = Sx0 + R * (Step1 + Step2 + Step3)

 Sy = Sy0 + (Height1 - 1) * R

 V1 = 1

 V2 = 1

 If Add1 = 0 Then

  Sx = Sx - R

  V1 = 0

  V2 = 0

 End If

 Call Cub(Sx, Sy, R, V1, V2, 1)

End Sub

Private Function Rand2(a1, a2)

' По 2 числам a1,a2 генерирует третье, не совпадающее с ними

 If a1 = a2 Then a2 = a2 + 1

L1000:

 R = 1 + Rand(a1 + a2)

 If R = a1 Then GoTo L1000

 If R = a2 Then GoTo L1000

 Rand2 = R

End Function

Private Sub Fixit(Mode, Tag, Answer1, Answer2, Answer3, Rulez, Answer)

'Создание отчета о неправильных ответах

 Static Title1 As String

 Static Good, NGood, a1, a2, A3

 Ans5 = Array("левый", "средний", "правый")

 Ans9 = Array("первое", "второе", "третье")

 Ans10 = Array("левая", "правая", "фигуры одинаковы")

 Ans = Val(Answer)

 If Mode = 0 Then 'Запомнить ответы

  NGood = Rulez

  a1 = Answer1

  a2 = Answer2

  A3 = Answer3

  Select Case Rulez

   Case 1

   Good = Answer1

   Case 2

   Good = Answer2

   Case 3

   Good = Answer3

  End Select

  Select Case Tag

  Case 1 To 3

   Title1 = "Вопрос " + Str(Tag) + _

    ". Сколько воды помещается в полой фигуре. Правильный ответ - " _

    + Str(Good) + ". Ваш ответ - "

  Case 4:

    Title1 = "Вопрос " + Str(Tag) + _

     ". Определение объема прямоугольного тетраэдра. Правильный ответ - " _

     + Str(Good) + ". Ваш ответ - "

  Case 5:

    Title1 = "Вопрос " + Str(Tag) + _

     ". Какой из чертежей соответствует условию задачи. Правильный ответ - " _

     + Ans5(NGood - 1) + ". Ваш ответ - "

  Case 6:

    Title1 = "Вопрос " + Str(Tag) + _

     ". Вычисление объема цилиндра. Правильный ответ - " _

     + Str(Good) + ". Ваш ответ - "

  Case 7:

   Title1 = "Вопрос " + Str(Tag) + _

   ". Вычисление объема конуса. Правильный ответ - " _

   + Str(Good) + ". Ваш ответ - "

  Case 8:

   Title1 = "Вопрос " + Str(Tag) + _

   ". Требовалось вычислить объем шара. Правильный ответ - " _

   + Str(Good) + ". Ваш ответ - "

  Case 9:

   Title1 = "Вопрос " + Str(Tag) + _

   ". Какое из условий задачи соответствует чертежу. Правильный ответ - " _

   + Ans9(NGood - 1) + ". Ваш ответ - "

  Case 10:

   Title1 = "Вопрос " + Str(Tag) + _

   ". Какая из фигур имеет больший объем. Правильный ответ - " _

   + Ans10(NGood - 1) + ". Ваш ответ - "

  End Select

 End If

 If Mode = 1 Then 'Сравнить с правильным

  If Ans <> NGood Then

   Select Case Ans

    Case 1

    A = a1

    Case 2

    A = a2

    Case 3

    A = A3

   End Select

   Select Case Tag

   Case 1 To 4:

    Title1 = Title1 + Str(A) + "." + Chr(13) + Chr(10)

   Case 5:

    Title1 = Title1 + Ans5(Ans - 1) + "." + Chr(13) + Chr(10)

   Case 6 To 8:

    Title1 = Title1 + Str(A) + "." + Chr(13) + Chr(10)

   Case 9:

    Title1 = Title1 + Ans9(Ans - 1) + "." + Chr(13) + Chr(10)

   Case 10:

    Title1 = Title1 + Ans10(Ans - 1) + "." + Chr(13) + Chr(10)

   End Select

   Text1.Text = Text1.Text + Title1

  End If

 End If

End Sub

Private Function Question(Tag)

'Функция формирования экранов для вопросов

 Dim Rulez As Integer 'Правильный ответ

 Dim Sx0, Sy0, Sx, Sy, R, V1, V2, V3 As Integer

 Dim Quest As String

 Dim Answer1, Answer2, Answer3 As Double

Option1.Value = False

Option2.Value = False

Option3.Value = False

Me.Cls

'If Tag = 0 Then Cool = 0

If Tag < 4 Then 'Первые 3 вопроса - про полые фигуры

 Quest = "В изображенную на картинке полую фигуру наливается вода." + _

  "Сколько кубических сантиметров воды поместится в фигуре, если она стоит на "

 Sx0 = 50

 Sy0 = 50

 R = 20

 Randomize

 Step1 = 1 + Rand(3)

 Height1 = 2 + Rand(2)

 Depth1 = 2 + Rand(2)

 Step2 = 1 + Rand(2)

 Step3 = 2 + Rand(2)

 Add1 = 1

 If Rand(3) = 0 Then Add1 = 0

 Call Fig(Sx0, Sy0, R, Height1, Depth1, Step1, Step2, Step3, Add1)

 'Добавляем отверстие

 Hx = Rand(Step2)

 Hy = Rand(Height1 - 1)

 Sx = Sx0 + Step1 * R + Hx * R

 Sy = Sy0 + Hy * R + R / 2

 Line (Sx, Sy)-Step(R / 2, R / 2), , BF

 If Tag = 1 Then

  Answer1 = (Height1 - 1 - Hy) * Depth1 * Step1 + (Depth1 - 1) * Step2 + _

   (Height1 - 1 - Hy) * (Step2 + Step3) + Add1 ' Правильный ответ

  Answer2 = Depth1 * Step1 * Height1 + Step2 * (Depth1 - 1) + _

   Step2 * Height1 + Height1 * Step3 + Add1

  Label1.Caption = Quest + "нижней (как на рисунке) грани?"

 ElseIf Tag = 2 Then 'Вопрос 2 про фигуру

  Answer1 = Height1 * Depth1 * Step1 + _

   Hx * (Height1 + Depth1 - 1) 'Правильный

  Answer2 = Height1 * Depth1 * Step1 * 2

  Label1.Caption = Quest + "ЛЕВОЙ грани?"

 ElseIf Tag = 3 Then 'Вопрос 3 про фигуру

  Answer1 = Step3 * Height1 + Add1 + (Step2 - 1 - Hx) * _

   (Height1 + Depth1 - 1) + (Height1 + Depth1 - 1) * 0.5 'Правильный

  Answer2 = (Height1 - 1 - Hy) * Depth1 * Step1 + (Depth1 - 1) * _

   Step2 + (Height1 - 1 - Hy) * (Step2 + Step3) + Add1

  Label1.Caption = Quest + "ПРАВОЙ грани?"

 End If

 Sx = Sx0

 Sy = Sy0 + 80

 CurrentX = Sx

 CurrentY = Sy

 Print "Объем 1 кубика равен 1 куб.см."

 Line (Sx, Sy + 15)-Step(R / 2, R / 2), , BF

 CurrentX = Sx + R

 CurrentY = Sy + 16

 Print " квадратное отверстие со стороной 0.5 см."

 If Answer2 = Answer1 Then Answer2 = Answer1 + 1

 Answer3 = Rand2(Answer1, Answer2)

 Rulez = Swap3(Answer1, Answer2, Answer3)

 Call Fixit(0, Tag, Answer1, Answer2, Answer3, Rulez, 0)

 Option1.Caption = Format(Answer1, ".0") + " куб.см."

 Option2.Caption = Format(Answer2, ".0") + " куб.см."

 Option3.Caption = Format(Answer3, ".0") + " куб.см."

ElseIf Tag = 4 Then 'тетраэдр

 Scal = 15

 A = 4 + Rand(5)

 B = 3 + Rand(5)

 C = 2 + Rand(7)

 Sx = 100

 Sy = 60

 Sb = Sqr((B * Scal) ^ 2 / 5)

 Sc = Sqr((C * Scal) ^ 2 / 5)

 'Линии из центра

 Line (Sx, Sy)-Step(0, A * Scal)

 Line (Sx, Sy)-Step(-2 * Sb, -Sb)

 Line (Sx, Sy)-Step(2 * Sc, -Sc)

 'Стороны

 Line (Sx, Sy + A * Scal)-(Sx - 2 * Sb, Sy - Sb)

 Line (Sx, Sy + A * Scal)-(Sx + 2 * Sc, Sy - Sc)

 Line (Sx - 2 * Sb, Sy - Sb)-(Sx + 2 * Sc, Sy - Sc)

 CurrentX = Sx

 CurrentY = Sy + 8

 Print "C"

 Answer1 = (A * B * C) / 2# 'Правильный

 Answer2 = A ^ 2 + B ^ 2 + C ^ 2

 Answer3 = Rand2(Answer1, Answer2)

 Rulez = Swap3(Answer1, Answer2, Answer3)

 Call Fixit(0, Tag, Answer1, Answer2, Answer3, Rulez, 0)

 Option1.Caption = Format(Answer1, ".0") + " куб.см."

 Option2.Caption = Format(Answer2, ".0") + " куб.см."

 Option3.Caption = Format(Answer3, ".0") + " куб.см."

 Label1.Caption = "Каков объем изображенного на рисунке тетраэдра," + _

  " если все углы при вершине С - прямые, а длины сторон равны " + _

  Str(A) + " , " + Str(B) + " , " + Str(C)

ElseIf Tag = 5 Then '3 пирамиды

 Dim x1, x2, x3 As Double

 n2 = 2 + Rand(5) 'Номер случайного чертежа - от 2 до 6

Line4:

 n3 = 2 + Rand(5)

 If n3 = n2 Then GoTo Line4

 x1 = 1

 x2 = n2

 x3 = n3

 Rulez = Swap3(x1, x2, x3)

 Call Fixit(0, Tag, Answer1, Answer2, Answer3, Rulez, 0)

 n1 = x1

 n2 = x2

 n3 = x3

 Call Pirr(n1, 55, 15)

 Call Pirr(n2, 155, 15)

 Call Pirr(n3, 255, 15)

 Option1.Caption = "Левый"

 Option2.Caption = "Средний"

 Option3.Caption = "Правый"

 Label1.Caption = "Задача формулируется так: 'В правильной" + _

 " треугольной пирамиде сторона основания равна a, а угол между" + _

 " смежными боковыми гранями равен d. Найдите объем пирамиды'." + _

 " Какой их этих чертежей соответствует условию?"

ElseIf Tag = 6 Then 'Цилиндр

 Sx = 100

 Sy = 30

 Scal = 8

 R = 3 + Rand(6)

 H = 6 + Rand(5)

 Circle (Sx, Sy), R * Scal, , , , 1 / 3

 Line (Sx + R * Scal, Sy)-Step(0, H * Scal)

 Line (Sx - R * Scal, Sy)-Step(0, H * Scal)

 Circle (Sx, Sy + H * Scal), R * Scal, , , , 1 / 3

 'Пунктирные линии

 DrawStyle = 1

 Line (Sx, Sy)-Step(0, H * Scal)

 Line (Sx, Sy + H * Scal)-Step(R * Scal, 0)

 DrawStyle = 0

 CurrentX = Sx + 8

 CurrentY = Sy + H * Scal

 Print "R"

 CurrentX = Sx + 8

 CurrentY = Sy + 8

 Print "H"

 Answer1 = R ^ 2 * H 'Правильный

 Answer2 = 6 * R * H

 Answer3 = Rand2(Answer1, Answer2)

 Rulez = Swap3(Answer1, Answer2, Answer3)

 Call Fixit(0, Tag, Answer1, Answer2, Answer3, Rulez, 0)

 Label1.Caption = "Каков объем изображенного на рисунке цилиндра," + _

 " если его радиус R равен " + Str(R) + " ,а высота H равна " + _

 Str(H) + "? Pi - это число Пи."

 Option1.Caption = Format(Answer1, ".0") + "*Pi куб.см."

 Option2.Caption = Format(Answer2, ".0") + "*Pi куб.см."

 Option3.Caption = Format(Answer3, ".0") + "*Pi куб.см."

ElseIf Tag = 7 Then 'Конус

 Sx = 100

 Sy = 15

 Scal = 10

Line2:

 R = 3 + Rand(4)

 H = R + 3 + Rand(5)

 If (H * R ^ 2) Mod 3 <> 0 Then GoTo Line2

 Circle (Sx, Sy + H * Scal), R * Scal, , , , 1 / 3

 Line (Sx, Sy)-(Sx + R * Scal, Sy + H * Scal)

 Line (Sx, Sy)-(Sx - R * Scal, Sy + H * Scal)

 DrawStyle = 1

 Line (Sx, Sy)-Step(0, H * Scal)

 Line (Sx, Sy + H * Scal)-Step(R * Scal, 0)

 DrawStyle = 0

 CurrentX = Sx + 8

 CurrentY = Sy + 10

 Print "H"

 CurrentX = Sx + 8

 CurrentY = Sy + H * Scal - 10

 Print "R"

 Answer1 = (R ^ 2 * H) / 3 'Правильный

 Answer2 = R ^ 2 * H ^ 2

 Answer3 = Rand2(Answer1, Answer2)

 Rulez = Swap3(Answer1, Answer2, Answer3)

 Call Fixit(0, Tag, Answer1, Answer2, Answer3, Rulez, 0)

 Label1.Caption = "Каков объем изображенного на рисунке конуса, если радиус" + _

 " основания R равен " + Str(R) + " ,а высота H равна " + Str(H) + _

 "? Pi обозначено число Пи."

 Option1.Caption = Format(Answer1, ".0") + "*Pi куб.см."

 Option2.Caption = Format(Answer2, ".0") + "*Pi куб.см."

 Option3.Caption = Format(Answer3, ".0") + "*Pi куб.см."

ElseIf Tag = 8 Then 'Шар

 Sx = 100

 Sy = 80

 Scal = 8

Line1:

 R = 3 + Rand(8)

 If (4 * R ^ 3) Mod 3 <> 0 Then GoTo Line1

 Circle (Sx, Sy), R * Scal

 Pi = 3.1415926

 Circle (Sx, Sy), R * Scal, , Pi, 2 * Pi, 1 / 3

 DrawStyle = 2

 Circle (Sx, Sy), R * Scal, , 0, Pi, 1 / 3

 DrawStyle = 1

 Line (Sx, Sy)-Step(R * Scal, 0)

 DrawStyle = 0

 CurrentX = Sx + 8

 CurrentY = Sy + 4

 Print "R"

 Answer1 = (4 * R ^ 3) / 3 'Правильный

 Answer2 = R ^ 3

 Answer3 = Rand2(Answer1, Answer2)

 Rulez = Swap3(Answer1, Answer2, Answer3)

 Call Fixit(0, Tag, Answer1, Answer2, Answer3, Rulez, 0)

 Label1.Caption = "Каков объем изображенного на рисунке шара, если его " + _

 "радиус R равен " + Str(R) + "? Pi - это число Пи."

 Option1.Caption = Format(Answer1, ".0") + "*Pi куб.см."

 Option2.Caption = Format(Answer2, ".0") + "*Pi куб.см."

 Option3.Caption = Format(Answer3, ".0") + "*Pi куб.см."

ElseIf Tag = 9 Then 'Какому из условий соответствует чертеж

 Array9 = Array("В правильной 4-угольной пирамиде боковые ребра наклонены к " + _

 "основанию под углом d=60 градусов, а длина бокового ребра равна 10 см." + _

 " Найти объем пирамиды.", "В правильной 4-угольной пирамиде плоский угол d" + _

 " при вершине равен 60 градусов, а длина бокового ребра равна 10 см. Найти объем пирамиды.", _

 "В правильной 3-угольной пирамиде боковые ребра наклонены к основанию под" + _

 " углом d=60 градусов, а длина бокового ребра равна 8 см. Найти объем пирамиды.", _

 "В правильной 4-угольной пирамиде плоский угол d между основанием и боковой " + _

 "поверхностью равен 60 градусов, а длина бокового ребра равна 10 см. Найти объем пирамиды.")

 Sx = 100

 Sy = 20

 Scal = 10

 Call MakePirr4(Sx, Sy, Sx - 4 * Scal, Sy + 11 * Scal, Sx + 3 * Scal, _

  Sy + 11 * Scal, Sx + 5 * Scal, Sy + 8 * Scal, Sx - 2 * Scal, Sy + 8 * Scal)

 DrawStyle = 2

 Line (Sx - 4 * Scal, Sy + 11 * Scal)-(Sx + 5 * Scal, Sy + 8 * Scal)

 x1 = Otn(Sx - 4 * Scal, Sx + 5 * Scal, 0.3)

 y1 = Otn(Sy + 11 * Scal, Sy + 8 * Scal, 0.3)

 x2 = Otn(Sx - 4 * Scal, Sx, 0.3)

 y2 = Otn(Sy + 11 * Scal, Sy, 0.3)

 DrawStyle = 1

 Line (x1, y1)-(x2, y2)

 DrawStyle = 0

 CurrentX = Sx + 3 * Scal

 CurrentY = Sy + 3 * Scal

 Print "10 см."

 CurrentX = Sx - 4 * Scal

 CurrentY = Sy + 11 * Scal + 4

 Print "d"

 n1 = 0

 n2 = 1 + Rand(3)

Line7:

 n3 = 1 + Rand(3)

 If n3 = n2 Then GoTo Line7

 Dim t1, t2, t3 As Double

 t1 = n1

 t2 = n2

 t3 = n3

 Rulez = Swap3(t1, t2, t3)

 Call Fixit(0, Tag, t1, t2, t3, Rulez, 0)

 n1 = t1

 n2 = t2

 n3 = t3

 Option1.Caption = Array9(n1)

 Option2.Caption = Array9(n2)

 Option3.Caption = Array9(n3)

 Label1.Caption = _

  "Какому из условий задачи соответствует данный чертеж?"

ElseIf Tag = 10 Then 'объем 2 фигур

 Step1 = 1 + Rand(2)

 Height1 = 2 + Rand(2)

 Depth1 = 2 + Rand(2)

 Step2 = 1 + Rand(2)

 Step3 = 2 + Rand(2)

 Add1 = 1

 If Rand(2) = 0 Then Add1 = 0

 Call Fig(30, 30, 15, Height1, Depth1, Step1, Step2, Step3, Add1)

 V1 = Height1 * Depth1 * Step1 + (Depth1 - 1) * Step2 + _

  Step2 * Height1 + Step3 * Height1 + Add1

 Step1 = 1 + Rand(2)

 Step2 = 1 + Rand(2)

 Step3 = 2 + Rand(2)

 Add2 = 1

 If Rand(2) = 0 Then Add2 = 0

 Call Fig(170, 30, 15, Height1, Depth1, Step1, Step2, Step3, Add2)

 V2 = Height1 * Depth1 * Step1 + (Depth1 - 1) * Step2 + _

  Step2 * Height1 + Step3 * Height1 + Add2

 Option1.Caption = "Левая фигура" '+ Str(V1)

 Option2.Caption = "Правая фигура" '+ Str(V2)

 Option3.Caption = "Объемы фигур одинаковы"

 Label1.Caption = "Какая из данных фигур имеет больший объем, если" + _

  " объем одного кубика равен 1 куб.см. и все кубики одинаковы по размеру?"

 If V1 > V2 Then Rulez = 1

 If V1 < V2 Then Rulez = 2

 If V1 = V2 Then Rulez = 3

 Call Fixit(0, Tag, V1, V2, V3, Rulez, 0)

End If

Question = Rulez

End Function

Private Sub Command1_Click() 'Щелчок по кнопке ответа

Static Tag As Integer 'Номер вопроса

Static Rulez As Integer

Static Cool As Integer 'Число правильных ответов

 If Tag > 0 Then

  Select Case Command1.Tag

   Case Rulez

    Cool = Cool + 1 'Ответ на пред. вопрос

   Case Else

    Call Fixit(1, Tag, 0, 0, 0, Rulez, Command1.Tag)

  End Select

 End If

 If Tag = 0 Then

  Cool = 0

  Text1.Visible = False

  Text1.Text = "Работа над ошибками:" + Chr(13) + Chr(10)

  Image1.Visible = False

  Image2.Visible = False

  Option1.Visible = True

  Option2.Visible = True

  Option3.Visible = True

  Label1.Visible = True

  Label2.Visible = False

  Me.ScaleTop = 0

  Me.ScaleLeft = 0

  Me.ScaleWidth = 400

  Me.ScaleHeight = 400

  AutoRedraw = -1 ' Свойство перерисовки графики

  Me.ForeColor = QBColor(15) ' Цвет линий = белый

  Me.BackColor = QBColor(1)  ' Цвет фона = синий

  Me.FillStyle = 1           ' заполнение

  Me.FillColor = QBColor(7) 'Цвет заполнения = серый

 End If

 If Tag < 11 Then

  Tag = Tag + 1

  Command1.Caption = "Ответ " + Str(Tag)

  Rulez = Question(Tag)

  'Command3.Caption = Str(Rulez) 'debug

  Command1.Enabled = False

 End If

 If Tag = 11 Then

  Tag = 0

  Command1.Caption = "Снова"

  Command1.Enabled = True

  Image2.Visible = True

  If Cool < 10 Then Text1.Visible = True

  Select Case Cool

   Case 0 To 7

    Label1.Caption = "К сожалению, Вы допустили более 2 ошибок при ответах" + _

     " на вопросы. Будьте внимательнее!"

   Case 8

    Label1.Caption = "Спасибо. У Вас 2 ошибки при ответе на вопросы теста." + _

     " Ваша оценка - ЧЕТЫРЕ с минусом."

   Case 9

    Label1.Caption = "Спасибо. Вы допустили только 1 ошибку при ответе на" + _

     " вопросы теста! Ваша оценка - ЧЕТЫРЕ."

   Case 10

    Label1.Caption = "Поздравляю! Вы правильно ответили на все вопросы теста" + _

     " и заслуживаете оценки ПЯТЬ!"

   Case Else

    Label1.Caption = "Программа не может оценить Ваши ответы!" + _

     " Обратитесь к разработчику..."

  End Select

  'Command3.Caption = Str(Cool) 'debug

  Option1.Visible = False

  Option2.Visible = False

  Option3.Visible = False

  Cool = 0

 End If

End Sub

Private Sub Command2_Click() 'Щелчок по кнопке О программе

 frmAbout.Text1.Text = "О программе" + Chr(13) + Chr(10) + _

 "Программа предназначена для проверки знаний учащихся по геометрии." + _

 " Проверка производится в форме тестирования - программа генерирует" + _

 " вопросы и предлагает учащимся ответить на них. " + Chr(13) + Chr(10) _

 + Chr(13) + Chr(10) + _

 "О тестах" + Chr(13) + Chr(10) + _

 "Геометрические фигуры и их размеры генерируются заново при каждом запуске теста!" + _

 " Расположение кнопки с правильным ответом по отношению к неправильным также случайно" + _

 " при любом ответе. Будьте внимательны при ответах, не надейтесь 'обмануть' программу!" _

 + Chr(13) + Chr(10) + Chr(13) + Chr(10) + _

 "Как выставляются оценки" + Chr(13) + Chr(10) + _

 "Если Вы правильно ответили на все вопросы, выставляется оценка '5'," + _

 " если допустили одну ошибку - оценка '4', за две ошибки программа ставит" + _

 " '4 с минусом', а в случае, если ошибок больше двух - предлагает попробовать " + _

 "свои силы еще раз... для этого предназначена кнопка 'Снова'" + Chr(13) + Chr(10) _

 + Chr(13) + Chr(10) + _

 "О работе над ошибками" + Chr(13) + Chr(10) + _

 "Если при ответах на вопросы теста были допущены ошибки - в верхней части окна появляется" + _

 " список ошибок с указанием Вашего и правильного варианта ответа. Это поможет Вам быть в " + _

 "следующий раз внимательнее и проверить правильность своих расчетов." + Chr(13) + Chr(10) + _

 Chr(13) + Chr(10) + _

 "Технические детали" + Chr(13) + Chr(10) + _

 "Программа написана на Visual Basic 5.0 с использованием только стандартных средств." + _

 " Проблемы с ее запуском могут быть связаны исключительно с отсутствием в системе нужных" + _

 " библиотек. Используя исходный текст программы как основу, будет нетрудно создать и другие тесты."

 frmAbout.Show vbModal, Me

End Sub

Private Sub Command3_Click() 'Щелчок по кнопке выхода

 Unload Me

End Sub

'Щелчки по радиокнопкам выбора ответа

Private Sub Option1_Click()

 Command1.Enabled = True

 Command1.Tag = 1

End Sub

Private Sub Option2_Click()

 Command1.Enabled = True

 Command1.Tag = 2

End Sub

Private Sub Option3_Click()

  Command1.Enabled = True

  Command1.Tag = 3

End Sub
Файл frmAbout.Frm

Option Explicit

' Reg Key Security Options...

Const READ_CONTROL = &H20000

Const KEY_QUERY_VALUE = &H1

Const KEY_SET_VALUE = &H2

Const KEY_CREATE_SUB_KEY = &H4

Const KEY_ENUMERATE_SUB_KEYS = &H8

Const KEY_NOTIFY = &H10

Const KEY_CREATE_LINK = &H20

Const KEY_ALL_ACCESS = KEY_QUERY_VALUE + KEY_SET_VALUE + _

                       KEY_CREATE_SUB_KEY + KEY_ENUMERATE_SUB_KEYS + _

                       KEY_NOTIFY + KEY_CREATE_LINK + READ_CONTROL

' Reg Key ROOT Types...

Const HKEY_LOCAL_MACHINE = &H80000002

Const ERROR_SUCCESS = 0

Const REG_SZ = 1                         ' Unicode nul terminated string

Const REG_DWORD = 4                      ' 32-bit number

Const gREGKEYSYSINFOLOC = "SOFTWARE\Microsoft\Shared Tools Location"

Const gREGVALSYSINFOLOC = "MSINFO"

Const gREGKEYSYSINFO = "SOFTWARE\Microsoft\Shared Tools\MSINFO"

Const gREGVALSYSINFO = "PATH"

Private Declare Function RegOpenKeyEx Lib "advapi32" Alias "RegOpenKeyExA" (ByVal hKey As Long, ByVal lpSubKey As String, ByVal ulOptions As Long, ByVal samDesired As Long, ByRef phkResult As Long) As Long

Private Declare Function RegQueryValueEx Lib "advapi32" Alias "RegQueryValueExA" (ByVal hKey As Long, ByVal lpValueName As String, ByVal lpReserved As Long, ByRef lpType As Long, ByVal lpData As String, ByRef lpcbData As Long) As Long

Private Declare Function RegCloseKey Lib "advapi32" (ByVal hKey As Long) As Long

Private Sub cmdSysInfo_Click()

  Call StartSysInfo

End Sub

Private Sub cmdOK_Click()

  Unload Me

End Sub

Public Sub StartSysInfo()

    On Error GoTo SysInfoErr

    Dim rc As Long

    Dim SysInfoPath As String

    ' Try To Get System Info Program Path\Name From Registry...

    If GetKeyValue(HKEY_LOCAL_MACHINE, gREGKEYSYSINFO, gREGVALSYSINFO, SysInfoPath) Then

    ' Try To Get System Info Program Path Only From Registry...

    ElseIf GetKeyValue(HKEY_LOCAL_MACHINE, gREGKEYSYSINFOLOC, gREGVALSYSINFOLOC, SysInfoPath) Then

        ' Validate Existance Of Known 32 Bit File Version

        If (Dir(SysInfoPath & "\MSINFO32.EXE") <> "") Then

            SysInfoPath = SysInfoPath & "\MSINFO32.EXE"

        ' Error - File Can Not Be Found...

        Else

            GoTo SysInfoErr

        End If

    ' Error - Registry Entry Can Not Be Found...

    Else

        GoTo SysInfoErr

    End If

    Call Shell(SysInfoPath, vbNormalFocus)

    Exit Sub

SysInfoErr:

    MsgBox "System Information Is Unavailable At This Time", vbOKOnly

End Sub

Public Function GetKeyValue(KeyRoot As Long, KeyName As String, SubKeyRef As String, ByRef KeyVal As String) As Boolean

    Dim i As Long                                           ' Loop Counter

    Dim rc As Long                                          ' Return Code

    Dim hKey As Long                                        ' Handle To An Open Registry Key

    Dim hDepth As Long                                      '

    Dim KeyValType As Long                                  ' Data Type Of A Registry Key

    Dim tmpVal As String                                    ' Tempory Storage For A Registry Key Value

    Dim KeyValSize As Long                                  ' Size Of Registry Key Variable

    '------------------------------------------------------------

    ' Open RegKey Under KeyRoot {HKEY_LOCAL_MACHINE...}

    '------------------------------------------------------------

    rc = RegOpenKeyEx(KeyRoot, KeyName, 0, KEY_ALL_ACCESS, hKey) ' Open Registry Key

    If (rc <> ERROR_SUCCESS) Then GoTo GetKeyError          ' Handle Error...

    tmpVal = String$(1024, 0)                             ' Allocate Variable Space

    KeyValSize = 1024                                       ' Mark Variable Size

    '------------------------------------------------------------

    ' Retrieve Registry Key Value...

    '------------------------------------------------------------

    rc = RegQueryValueEx(hKey, SubKeyRef, 0, _

                         KeyValType, tmpVal, KeyValSize)    ' Get/Create Key Value

    If (rc <> ERROR_SUCCESS) Then GoTo GetKeyError          ' Handle Errors

    If (Asc(Mid(tmpVal, KeyValSize, 1)) = 0) Then           ' Win95 Adds Null Terminated String...

        tmpVal = Left(tmpVal, KeyValSize - 1)               ' Null Found, Extract From String

    Else                                                    ' WinNT Does NOT Null Terminate String...

        tmpVal = Left(tmpVal, KeyValSize)                   ' Null Not Found, Extract String Only

    End If

    '------------------------------------------------------------

    ' Determine Key Value Type For Conversion...

    '------------------------------------------------------------

    Select Case KeyValType                                  ' Search Data Types...

    Case REG_SZ                                             ' String Registry Key Data Type

        KeyVal = tmpVal                                     ' Copy String Value

    Case REG_DWORD                                          ' Double Word Registry Key Data Type

        For i = Len(tmpVal) To 1 Step -1                    ' Convert Each Bit

            KeyVal = KeyVal + Hex(Asc(Mid(tmpVal, i, 1)))   ' Build Value Char. By Char.

        Next

        KeyVal = Format$("&h" + KeyVal)                     ' Convert Double Word To String

    End Select

    GetKeyValue = True                                      ' Return Success

    rc = RegCloseKey(hKey)                                  ' Close Registry Key

    Exit Function                                           ' Exit

GetKeyError:      ' Cleanup After An Error Has Occured...

    KeyVal = ""                                             ' Set Return Val To Empty String

    GetKeyValue = False                                     ' Return Failure

    rc = RegCloseKey(hKey)                                  ' Close Registry Key

End Function

Вид окна программы при запуске

Вид окна программы после выставления оценки

